

- [All Categories](#)

Search the Adafruit Learning

- [Home/](#)
- [Sensors/](#)
- [Temperature/](#)
- [Thermistor/](#)
- Using a Thermistor

## Featured in this Guide

10K Precision Epoxy Thermistor



\$4.00 [Add to Cart](#)

[Add all to Cart](#)

[×](#)

### Out of Stock Notification

Your Name:  Your E-mail:

[Notify Me](#)


## Using a Thermistor Created by [Ladyada](#) **Thermistor** Connecting to a Thermistor

[Measure temperature using a resistor!](#)

These thermistors are pretty hardy, you can strip the PVC insulation and stick the wires into a breadboard or solder to them directly. Otherwise you can cut or extend the wires. Since the resistance is pretty high (10Kohm) the wire resistance won't make a huge difference.

- [Overview](#)
- [Using a Thermistor](#)
- [Buy a Thermistor](#)
- [Single Page](#)
- [Download PDF](#)

[Feedback? Corrections?](#)



## Analog Voltage Reading Method

To measure the temperature, we need to measure the resistance. However, a microcontroller does not have a resistance-meter built in. Instead, it only has a voltage reader known as a analog-digital-converter. So what we have to do is convert the resistance into a voltage, and we'll do that by adding another resistor and connecting them in series. Now you just measure the voltage in the middle, as the resistance changes, the voltage changes too, according to the simple voltage-divider equation. We just need to keep one resistor fixed

Say the fixed resistor is **10K** and the variable resistor is called **R** - the voltage output (**V<sub>o</sub>**) is:

$$V_o = R / (R + 10K) * V_{cc}$$

Where **V<sub>cc</sub>** is the power supply voltage (3.3V or 5V)

Now we want to connect it up to a microcontroller. Remember that when you measure a voltage (**V<sub>i</sub>**) into an Arduino ADC, you'll get a number.

$$\text{ADC value} = V_i * 1023 / V_{cc}$$

So now we combine the two (**V<sub>o</sub> = V<sub>i</sub>**) and get:

$$\text{ADC value} = R / (R + 10K) * V_{cc} * 1023 / V_{cc}$$

What is nice is that if you notice, the **V<sub>cc</sub>** value cancels out!


$$\text{ADC value} = R / (R + 10K) * 1023$$

It doesn't matter what voltage you're running under. Handy!

Finally, what we really want to do is get that **R** (the unknown resistance). So we do a little math to move the **R** to one side:

$$R = 10K / (1023/ADC - 1)$$

Great, let's try it out. Connect up the thermistor as shown:



Connect one end of the 10K resistor to 5V, connect the other end of the 10K 1% resistor to one pin of the thermistor and the other pin of the thermistor to ground. Then connect Analog 0 pin to the 'center' of the two.

Now run the following sketch:

```
Copy Code
1. // the value of the 'other' resistor
2. #define SERIESRESISTOR 10000
3.
4. // What pin to connect the sensor to
5. #define THERMISTORPIN A0
6.
7. void setup(void) {
8.   Serial.begin(9600);
9. }
10.
11. void loop(void) {
12.   float reading;
13.
14.   reading = analogRead(THERMISTORPIN);
15.
16.   Serial.print("Analog reading ");
17.   Serial.println(reading);
18.
19.   // convert the value to resistance
20.   reading = (1023 / reading) - 1;
21.   reading = SERIESRESISTOR / reading;
22.   Serial.print("Thermistor resistance ");
23.   Serial.println(reading);
24.
25.   delay(1000);
26. }
```

You should get responses that correspond to the resistance of the thermistor as measured with a multimeter


## Better Readings

When doing analog readings, especially with a 'noisy' board like the arduino, we suggest two tricks to improve results. One is to use the 3.3V voltage pin as an analog reference and the other is to take a bunch of readings in a row and average them.

The first trick relies on the fact that the 5V power supply that comes straight from your computer's USB does a lot of stuff on the Arduino, and is almost always much noisier than the 3.3V line (which goes through a secondary filter/regulator stage!) It's easy to use, simply connect 3.3V to AREF and use that as the VCC voltage. Because our calculations don't include the VCC voltage, you don't have to change your equation. You do have to set the analog reference but that's a single line of code

Taking multiple readings to average out the result helps get slightly better results as well, since you may have noise or fluctuations, we suggest about 5 samples.

Rewire as shown:



This sketch takes those two improvements and integrates them into the demo, you will have better, more precise readings.

```
Copy Code
1. // which analog pin to connect
2. #define THERMISTORPIN A0
3. // how many samples to take and average, more takes longer
4. // but is more 'smooth'
5. #define NUMSAMPLES 5
6. // the value of the 'other' resistor
7. #define SERIESRESISTOR 10000
8.
9. int samples[NUMSAMPLES];
10.
11. void setup(void) {
12.   Serial.begin(9600);
13.   // connect AREF to 3.3V and use that as VCC, less noisy!
14.   analogReference(EXTERNAL);
15. }
16.
17. void loop(void) {
18.   uint8_t i;
19.   float average;
20.
21.   // take N samples in a row, with a slight delay
22.   for (i=0; i< NUMSAMPLES; i++) {
23.     samples[i] = analogRead(THERMISTORPIN);
24.     delay(10);
25.   }
26.
27.   // average all the samples out
28.   average = 0;
29.   for (i=0; i< NUMSAMPLES; i++) {
30.     average += samples[i];
31.   }
32.   average /= NUMSAMPLES;
33.
34.   Serial.print("Average analog reading ");
35.   Serial.println(average);
36.   // convert the value to resistance
37.   average = 1023 / average - 1;
```

```

38.     average = SERIESRESISTOR / average;
39.
40.     Serial.print("Thermistor resistance ");
41.     Serial.println(average);
42.
43.     delay(1000);
44. }

```

## Converting to Temperature

Finally, of course, we want to have the temperature reading, not just a resistance! If you just need to do a quick comparison circuit (if temperature is below X do this, if its above Y do that), you can simply use the temperature/resistance table which correlates the resistance of the thermistor to the temperature.

However, you probably want actual temperature values. [To do that we'll use the Steinhart-Hart equation](#), which lets us do a good approximation of converting values. Its not as exact as the thermistor table (it is an approximation) but its pretty good around the temperatures that this thermistor is used.

$$\frac{1}{T} = A + B \ln(R) + C(\ln(R))^3$$

However, this equation is fairly complex, and requires knowing a lot of variables that we don't have for this thermistor. [Instead we will use the simplified B parameter equation.](#)

$$\frac{1}{T} = \frac{1}{T_0} + \frac{1}{B} \ln\left(\frac{R}{R_0}\right)$$

For this one we only need to know **T<sub>0</sub>** (which is room temperature, 25 °C = 298.15 K) **B** (in this case 3950, the coefficient of the thermistor), and **R<sub>0</sub>** (the resistance at room temp, in this case 10Kohm). We plug in **R** (resistance measured) and get out **T** (temperature in Kelvin) which is easy to convert to °C

The following sketch will calculate °C for you

```

Copy Code
1. // which analog pin to connect
2. #define THERMISTORPIN A0
3. // resistance at 25 degrees C
4. #define THERMISTORNOMINAL 10000
5. // temp. for nominal resistance (almost always 25 C)
6. #define TEMPERATURENOMINAL 25
7. // how many samples to take and average, more takes longer
8. // but is more 'smooth'
9. #define NUMSAMPLES 5
10. // The beta coefficient of the thermistor (usually 3000-4000)
11. #define BCOEFFICIENT 3950
12. // the value of the 'other' resistor
13. #define SERIESRESISTOR 10000
14.
15. int samples[NUMSAMPLES];
16.
17. void setup(void) {
18.     Serial.begin(9600);
19.     analogReference(EXTERNAL);
20. }
21.
22. void loop(void) {
23.     uint8_t i;
24.     float average;
25.
26.     // take N samples in a row, with a slight delay
27.     for (i=0; i< NUMSAMPLES; i++) {
28.         samples[i] = analogRead(THERMISTORPIN);
29.         delay(10);
30.     }
31.
32.     // average all the samples out
33.     average = 0;
34.     for (i=0; i< NUMSAMPLES; i++) {
35.         average += samples[i];
36.     }
37.     average /= NUMSAMPLES;
38.


```



```

39. Serial.print("Average analog reading ");
40. Serial.println(average);
41.
42. // convert the value to resistance
43. average = 1023 / average - 1;
44. average = SERIESRESISTOR / average;
45. Serial.print("Thermistor resistance ");
46. Serial.println(average);
47.
48. float steinhart;
49. steinhart = average / THERMISTORNOMINAL; // (R/Ro)
50. steinhart = log(steinhart); // ln(R/Ro)
51. steinhart /= BC0EFFICIENT; // 1/B * ln(R/Ro)
52. steinhart += 1.0 / (TEMPERATURENOMINAL + 273.15); // + (1/To)
53. steinhart = 1.0 / steinhart; // Invert
54. steinhart -= 273.15; // convert to C
55.
56. Serial.print("Temperature ");
57. Serial.print(steinhart);
58. Serial.println(" *C");
59.
60. delay(1000);
61. }

```



For better precision, we suggest reading the exact value of the 'series 10K' it should be nearly exactly 10K but if you can get a better reading that will reduce your error.

## How Precise is the Reading?

You may notice that above, the temperature reading is 28.16°C - does that mean we have 0.01°C precision? Unfortunately no! The thermistor has error and the analog reading circuitry has error.

We can approximate the expected error by first taking into account the thermistor resistance error. The thermistor is correct to 1%, which means that at 25°C it can read 10,100 to 9900 ohms. At around 25°C a difference of 450 ohms represents 1°C so 1% error means about  $\pm 0.25^{\circ}\text{C}$  (you may be able to calibrate this away by determining the resistance of the thermistor in a 0°C ice bath and removing any offset). You can also spring for a 0.1% thermistor which will reduce the possible resistance error down to  $\pm 0.03^{\circ}\text{C}$


Then there is the error of the ADC, for every bit that it is wrong the resistance (around 25°C) can be off by about 50 ohms. This isn't too bad, and is a smaller error than the thermistor error itself  $\pm (0.1^{\circ}\text{C})$  but there is no way to calibrate it 'away' - a higher precision ADC (12-16 bits instead of 10) will give you more precise readings

In general, we think thermistors are higher precision than thermocouples, or most low cost digital sensors, but you will not get better than  $\pm 0.1^{\circ}\text{C}$  accuracy on an Arduino with a 1% thermistor and we would suggest assuming no better than  $\pm 0.5^{\circ}\text{C}$ .

[< Testing a Thermistor](#)

Last updated on 2012-07-02 at 03.38.54 PM

## Related Guides



•

### [USB, DC & Solar Lipoly Charger](#)

[Power your projects from the sun!](#)



•

### [Basic Resistor Sensor Reading on Raspberry Pi](#)

[Reading resistive sensors with RC timing](#)

## **About Adafruit**

Adafruit was founded in 2005 by MIT engineer, Limor "Ladyada" Fried. Her goal was to create the best place online for learning electronics and making the best designed products for makers of all ages and skill levels.

[Learn More.](#)

## **The Adafruit Learning System**

In 2012, after years of coding tutorials by hand, Limor put together a small team to build a custom tutorial management system from the ground up. The Adafruit Learning System allows us to make a wide range of awesome tutorials fast and efficiently. We hope these tutorials will help you learn something new, and inspire you to make something great!

## **Useful Stuff**

[Support Forums](#) [Adafruit Shop](#) [RSS](#)

327 tutorials and counting



Powered by Adafruit Learning Technologies